第二讲　段及数据的定义

一、段的完整定义（P80: 3.3.4）

定义不同作用的段（每个段对应一片存储区，用段寄存器存放存储区的首地址）

```
#define  TRUE  1
#define  FALSE  0
main()
{
   int sum=0; //初始化变量
   char flag=TRUE;
   int i;



   for(i=1;i<11;i++)
   {
       sum = sum+i;//求和
       if(sum==20)
       {
            goto Over;
         }
     }
Over: flag=FALSE;
}
```

```
.386
DUIZHAN SEGMENT USE16 STACK ; 堆栈段
          DB 200 DUP(0)
DUIZHAN  ENDS
DATA SEGMENT USE16  ; 数据段
  sum   dw 0            ; 初始化变量
  TRUE  = 1
  FALSE EQU 0
  flag db TRUE
  i    dw ?
DATA ENDS
CODE SEGMENT  USE16  ; 代码段
      ASSUME DS:DATA,CS:CODE, SS: DUIZHAN
main proc far
start:   mov ax,DATA ; 数据段首址→AX(MOV 为传送指令)
         mov ds,ax   ; 数据段首址 DATA 置入数据段寄存器 DS
         mov i, 1    ; i=1, 计数器赋初值
         mov ax,sum
next:     add ax,i      ; 求和
         cmp ax,20   ; 和值是否等于 20
         je   Over
         inc i       ; i++,计数器加 1
         cmp i,11    ; 计数值是否达到 11
         jb   next
Over:    MOV sum,AX  ; 保存和值
         MOV flag,FALSE  ; 设定标志值
main     endp
CODE       ENDS
               END start
```

（了解 段的简化定义：P240）

二、数据的定义(P47：3.1)

1．常量与数值表达式

① 常量

定义：常量是指在汇编时已有确定的数值。

常量可分为：

数值常量；1，2AH，-200 。。。

符号常量：使用等价伪指令语句“EQU”或等号伪指令语句“=”定义，定义后直接引用符号名（符号名长度控制在<=31 个字节）；

它不分配存储单元，只建立等价代换关系；

```
TRUE   EQU   1                                    (C 语言里？)
FALSE    =   0
```

可出现在任何段中。

②数值表达式

定义：由常量（数值常量、符号常量、标号或变量）及运算符组成的有确定意义的式子。

例如： MOV AX, (TRUE+1)/2　　X-Y， OFFSET X

运算符：十、一、*、/ 、移位、逻辑运算等，参见 P48。

（数值运算符与运算程序的区别）

2. 标号和变量

（1）标号

含义：在代码段中由用户定义的、以冒号作结束的符号为标号（标号名长度<=31 个字节）。它表示了该指令的偏移地址；

标号的类型：

NEAR（近），本段内使用。

FAR（远），远标号则指调用其它段中所定义的标号。

```
例     Over:      MOV sum,AX
```

（2）变量

含义：变量是数据段中一个数据存储单元的名字（变量名<=31 个字节）。

可代表一批存储单元的首址。

变量的类型：（每个）数据项存储单元的大小。

如 BYTE（字节）、　　伪指令 DB 在定义变量时指出

WORD（字）、　　伪指令 DW

DWORD（双字）、　　伪指令 DD

FWORD（3 个字）、　　伪指令 DF

QWORD（4 个字）、　　伪指令 DQ

TBYTE（10 个字节），　伪指令 DT。

格式：　[变量名]　DX　表达式[，表达式]

功能：定义了一变量，并开辟了由变量属性所决定的一片连续存储区，其存储区所占字节数=表达式个数*变量的类型。

例如：BUF　DB　‘How Are You，ABCD12EF……’，20H　；BUF 的类型为字节

ARR　DW　10，-60，189，‘AB’，‘C’　；ARR 的类型为字

TT　DD　0A57BD36H　；TT 的类型为双字

变量表达式的形式可为：

① 数值表达式；　　含*变量名*

② ？　表示只分配存储单元，不需要置初值；

③ ASCII 串；

④ n　DUP(?)　其中 n>0

⑤ n　DUP (表达式　[，表达式])

例如：　DB　3　DUP（’A’,’B’）

展开为 DB　‘A’, ’B’, ’A’, ’B’, ’A’, ’B’

⑥ n　DUP (m　DUP(表达式 1，…，表达式 x)，表达式)；

例如：　DB　4　DUP（3 DUP　（2），3）

展开为：　DB　2，2，2，3，2，2，2，3，2，2，2，3，2，2，2，3

⑦ 其上表达式的组合。

如：DB　‘ABC’，？，10　DUP（0），0AH, 0DH

小结：

标号和变量都是用户自己定义的、代表数据/代码存放地址的符号。

一般地：标号在代码段，变量在数据段。

用户定义的标号/变量都有以下三个属性：

- 段地址：标号/变量所在段的段首址。SEG　<变量或标号>

分离出其后变量或标号的段首址。

当引用该标号/变量时，该段首址应在某一段寄存器中，此时该段称为 CPU 当前可访问段；

- 偏移地址：该标号/变量到段首址的字节距离，为 16/32 位无符号数；

  OFFSET <变量或标号或表达式>

  分离出其后变量或标号或表达式的偏移地址。

- 类型。（标号的远近、变量的字长）

  可以用类型运算符 **PTR** 来临时指定地址类型。如：

  MOV BYTE PTR DS:[100H]，5

对变量的访问：

```
DATA   SEGMENT USE16
flag   DB ’ AB’ ,20H, “I’ m”
sum    DW 1234H, ’ AB’
TRUE   EQU  1
AOH    DW  sum ,OAH+ TRUE
A2     DW  $-AOH ;($=A2 的 EA,
                  即 A2-AO)
A3     DD  sum
DATA   ENDS
```

汇编地址计数器 $

（内存编址，字、双字的地址）

```
MOV    AL, flag
MOV    AL, flag+1
MOV    AX, SEG flag

MOV    AX, sum
MOV    AX, sum+1
```

		xx	1000H
		xx	1001H
flag	0	41H	DATA=1002H
	1	42H	
	2	20H	
	3	49H	
	4	27H	
	5	6DH	
sum	6	34H	
	7	12H	
	8	42H	
	9	41H	
AOH	10	06H	
	11	00H	
	12	0BH	
	13	00H	
A2	14	04H	
	15	00H	
A3	16	06H	
	17	00H	
	18	02H	
	19	10H	

结构：P246

（1）结构说明的一般格式如下：

结构名　**STRUCT**

数据定义语句序列

结构名　**ENDS**

课程的结构 **COURSE** 的说明：

```
COURSE   STRUCT
         CID          DD ?                        ；课程编号
         CTITLE       DB  20 DUP（0）             ；课程名
         CHOUR        DB  0                       ；学时数
         CTEACHER     DB  ‘WANG ZHONGLING ’       ；主讲教师
         CTERM        DB  1，2                    ；开课学期
COURSE   ENDS
```

结构说明应放在结构变量定义之前，不属于任何段。

（2）结构变量定义的一般格式为：

变量名　结构名　<字段赋值表 >

例如：

```
C1   COURSE  < >              ；5 个字段均用结构说明时给的初值
C2   COURSE  <2102，’SHU XUE’，60，’LI MING’，> ；仅 CTERM 字段未重新赋值
     COURSE  <2103，’YU WEN’，80，，>   ；CTEACHER 和 CTERM 字段未重新赋值，省略了变量名
C4   COURSE  5 DUP(<2101，，40，，>)     ；定义了 5 个相同的结构变量，对 CID、CHOUR 重新赋了值
     COURSE  10 DUP(<>)          ；定义了 10 个结构变量，即预留了相应的存储空间
```

（3）访问。

a)“结构变量名.结构字段名”的形式。例如：

```
.DATA
  C2   COURSE  <2102，’SHU XUE’，60，’LI MING’，1,2>
…
.CODE
        MOV   EAX，C2.CID          ；将 2102 送到 EAX 寄存器中
        MOV   AL，C2.CTITLE        ；CTITLE 中的字符‘S’送到 AL 中
        MOV   AH，C2.CTITLE+2      ；CTITLE 中的字符‘U’送到 AH 中
```

b)其他形式

}